PLC Modification of Lubrication System for Mining Dump Trucks

Shenglin Ding, Min Yu, Zhizhan Zhou
Dexing Copper Mine, Jiangxi Copper Corporation Limied, Dexing, Jiangxi, China

Abstract: The automatic lubrication system of the mining dump truck is provided with pressure by a hydraulic gear pump. The pressure sensor provides real-time working pressure feedback for the lubrication system, serving as a basis for the electromagnetic pressure relief valve control system. The lubrication condition of the articulated bearings in the mining dump truck has a significant impact on the truck's utilization rate. Proper lubrication is directly related to extending equipment lifespan and reducing equipment maintenance costs. By researching and analyzing the principles of the automatic lubrication control system for mining dump trucks, and considering the characteristics, types, and maintenance difficulties of lubrication system failures, we propose a PLC modification to the original control system. This modification aims to enhance system operational stability and reduce maintenance costs.

Keywords: Automatic Lubrication Control System; PLC; Stability; Input-Output Nodes

1. Introduction

As the main equipment produced by Dexing copper mine, 730E, 830E, NTE200 and other electric wheel dump trucks undertake the production tasks of mining and soil removal and transportation in the mine. Their lubrication system is an independent system separated from the truck control (Zhang, 2013), and the working condition of the lubrication system is an important factor restricting the utilization rate of electric wheels, and the automatic lubrication control system of the vehicle has problems such as high unit price of spare parts, long procurement cycle, and inconvenient lubrication fault maintenance, which greatly affects the maintenance efficiency of the system.

2. Original Truck Lubrication System

2.1 System Composition

Oil storage tank, hydraulic motor, front pressure switch, rear pressure switch, flow control valve, solenoid valve, pressure reducing valve, main oil pipeline, oil injector and other components[1].

2.2 Control Loops

Figure 1 is the 830E electric wheel automatic lubrication electrical control circuit diagram, the working principle is that during the operation of the truck (hydraulic system establishes the working pressure), the control time set by the solid-state lubrication timer is connected to the power supply to the solenoid valve, the solenoid valve opens the oil supply of the hydraulic motor, provides the output of high-pressure grease that drives the Lincoln lubrication pump, and monitors the output of grease by the rear and front pressure switches to complete a lubrication cycle.



Figure 1. Automatic Lubrication of the Control Loop

2.3 Difficulties of System Maintenance

First, the working condition of the pressure switch in the lubrication control circuit has a great impact on the operating stability of the system, and the change of the working state of the pressure switch (when the set pressure value is reached, the switch state transition) is difficult to monitor in real time during the actual maintenance process on site; There are too many input and output nodes such as control loop repeaters and pressure switches [2-4], which increases the difficulty of troubleshooting lubrication system failures.

Second, the solid-state lubrication control timer is an imported part, with high unit price and long procurement cycle, especially due to the impact of the epidemic in recent years, and procurement is difficult.

3. PLC Modification

In order to reduce the cost of equipment operation and maintenance, reduce the equipment failure rate, and improve the troubleshooting efficiency of the lubrication system, the technicians decided to carry out technical transformation of the lubrication control system of 730E, 830E and NTE200 electric wheels, using Siemens S7-200PLC, combined with the characteristics and working requirements of electric wheel truck lubrication control, independently wrote the lubrication control program, and then after component, line installation and program debugging, the test machine was successful [5]. Through the tracking and observation of the operation of the equipment, the lubrication effect of the modified equipment is good.

3.1 PLC Address Allocation

Table 1 describes the modified lubrication control I/O (input and output) points.

Table 1. The Modification of PLC Addresses Assignment

Input		Output	
I0.0	Manual Lubrication Switch	Q0.0	Lubrication Solenoid Valve Relay
I0.1	Lubrication Pressure Switch		
I0.2	Lube Oil Level Switch	Q0.1	Lubrication Fault Lamp Relay
I0.3	Start Signal		

3.2 Control Requirements

3.2.1 The lubrication control system can only work automatically or be tested manually if the lubricating oil tank is in an oil state (oil level switch I0.2 closed) and the vehicle starts (start signal I0.3 closed).

3.2.2 The lubrication oil storage barrel is in the state of oil shortage (the lubricating oil level switch I0.2 is open), and the lubrication fault lamp relay has an electric action alarm until the lubricated oil storage barrel returns to the oily state (the lubricating oil level switch I0.2 is closed).

3.2.3 Lubricating oil storage barrel has oil state (lubricating oil level switch I0.2 closed), start the vehicle (start signal I0.3 closed), start 15min timing, when the timing time arrives (or press the manual test switch I0.0 during the timing time), the lubrication solenoid valve relay (Q0.0) gets electrified action, and controls the Lincoln pump to start working.

3.2.4 Within 3 minutes of the lubrication solenoid valve relay getting power, if the lubrication pressure switch (I0.1) is active, the lubrication solenoid valve relay is powered off, and the system automatically enters the next working cycle.

3.2.5 Within 3min of the lubrication solenoid valve relay, if the lubrication pressure switch (I0.1) does not operate, then after the end of the 3min timekeeping, the lubrication solenoid valve relay is powered off, and the lubrication fault lamp relay (Q0.1) is powered off alarm, and the system automatically enters the next working cycle, but the lubrication fault light will remain on [6, 7] until the lubrication pressure switch acts or the vehicle turns off and the power is cut off.

3.3 Advantages of Maintenance After PLC Modification

After the transformation of the lubrication control system, the input and output status of lubricating oil level, grease pressure, solenoid valve and so on can be visualized, and the approximate cause of the fault can be easily and quickly judged when the lubrication fault is overserved; At the same time, the timing adjustment is fast, and the oil supply output can be flexibly adjusted according to the lubrication condition and seasonal changes, so that the lubrication system can not only ensure reliable lubrication of the equipment, but also achieve the purpose of reducing grease consumption [8-10].

3.4 PLC Control Program

3.4.1 Program 1: Sequence control design method

Figure 2 to Figure 6 is PLC ladder diagram compiled by sequential control method, and Figure 7 is PLC flow chart of sequence control method. Code samples are shown in Appendix A.



Figure 2. Sequence Control Method Design: Ladder Chart 1



Figure 3. Sequence Control Method Design: Ladder Chart 2



Figure 4. Sequence Control Method Design: Ladder Chart 3



Figure 5. Sequence Control Method Design: Ladder Chart 4



Figure 6. Sequence Control Method Design: Ladder Chart 5



Figure 7. Sequence Control Method Design: Flow Chart

3.4.2 Program 2: Empirical Design Method

Figure 8 and Figure 9 shows a PLC ladder compiled empirically. Code samples are shown in Appendix B.



Figure 8. Empirical Design Method: Ladder Chart 1



Figure 9. Empirical Design Method: Ladder Chart 2

3.4.3 PLC Wiring Diagram

Figure 10 depicts the PLC external wiring corresponding to the original system to implement the set control function.



Figure 10. PLC Wiring Diagram

4. Conclusion

Through the PLC transformation of the original lubrication control system of the electric wheel self-loading truck, the operation stability of the lubrication system is guaranteed; The failure rate of the lubrication system is significantly reduced, and the maintenance efficiency of the lubrication system fault is greatly improved. The impact of imported spare parts procurement factors on equipment operation is alleviated.

Appendix A. Sequence Control Method Code Samples

TITLE=PROGRAM COMMENTS

Network 1 // NETWORK TITLE

// Initiation Step

```
LD      SM0.1
ON      I0.2
ON      I0.3
S       M0.0, 1
R       M0.1, 10
Network 2
//      Lubrication Waiting Step
LD      M0.0
A       I0.2
A       I0.3
LD      M0.3
A       T39
OLD
LD      M0.4
A       T40
OLD
S       M0.1, 1
R       M0.0, 1
R       M0.3, 1
R       M0.4, 1
Network 3
//      Lubrication Working Step
LD      M0.1
LD      T37
O       I0.0
ALD
S       M0.2, 1
R       M0.1, 1
Network 4
//      Lubrication Done, Normal, Interim Step
LD      M0.2
A       I0.1
S       M0.3, 1
R       M0.2, 1
Network 5
//      Lubrication Done, Error, Interim Step
LD      M0.2
A       T38
S       M0.4, 1
R       M0.2, 1
Network 6
//      Lubrication Waiting Cycle Timing
LD      M0.1
TON     T37, 9000
Network 7
//      Lubrication Solenoid Valve Relay Controller and Lubrication Working Cycle Timing
LD      M0.2
=       Q0.0
TON     T38, 1800
Network 8
//      Lubrication Normal, Interim Timing
LD      M0.3
TON     T39, 2
Network 9
//      Lubrication Error, Interim Timing
LD      M0.4
TON     T40, 2
Network 10
//      Lubrication Error Self Locking
LD      M0.4
O       M1.0
AN      I0.1
=       M1.0
Network 11 //    NETWORK TITLE
//      Lubrication Fault Lamp Relay Controller
LD      M1.0
ON      I0.2
=       Q0.1
```

Appendix B. Empirical Design Method Code Samples

```
TITLE=PROGRAM COMMENTS
Network 1 //    NETWORK TITLE
//      Network Comments
LD      I0.2
A       I0.3
LPS
AN      Q0.0
TON     T37, 9000
LPP
LD      T37
O       I0.0
O       Q0.0
AN      M0.0
AN      M0.1
ALD
=       Q0.0
TON     T38, 1800
Network 2
LD      Q0.0
LPS
A       I0.1
=       M0.0
LPP
A       T38
=       M0.1
Network 3
LD      M0.1
O       M1.0
AN      I0.1
=       M1.0
Network 4
LD      M1.0
ON      I0.2
=       Q0.1
```


References

[1] Zhang, W.Q. (2013). Investigation and Treatment of Common Faults of Automatic Lubrication System of Large Mining Electric Wheel. China Mining Engineering, 42(04), 51-54.
[2] Chang, L., & He, J.Q. (2022). Mechanical Grab Non-powered Automatic Lubrication System. Port Technology, (12), 35-58.
[3] Xie, N.F. (2002). Application of Automatic Lubrication of Oil Injectors and System Control Loops. Mechanical Research and Application, (01), 18-19.
[4] Lv, J.H., & Zhou, Y.H. (2021). Common Fault Diagnosis and Treatment of Electric Wheel Automobile Automatic Lubrication System. Copper Engineering, (01), 89-91.
[5] Cao, X.Y. (2011). Design and Research of Automatic Lubrication System of 125MN Extruders. Heavy Machinery, (03), 52-54.
[6] Liu, W. (2021). MT5500 Truck Automatic Lubrication System Fault Analysis and Troubleshooting Method. Inner Mongolia Coal Economy, (03), 164-165.
[7] Cheng, J., & Liu, B. (2020). Design of Energy-saving Control Method for Automatic Lubrication System of CNC Machine Tool. Machine Tools and Hydraulics, 48(11), 125-127.
[8] Hou, Q.P. (2019). Innovations in Centralized Automatic Lubrication Systems at the Station. Petrochemical Technology, 26(12), 91-93.
[9] Chen, L., & Gao, X.B. (2017). PLC-based Conveyor Chain Automatic Lubrication Control System. Equipment Management and Maintenance, (01), 91-93.
[10] Li, F., & Lu, Y.C. (2016). Failure Analysis of the Failure of the 930E Truck's Automatic Lubrication System. Surface Mining Technology, 31(03), 78-80.